api_sync (body)

CREATE OR REPLACE PACKAGE BODY api_sync
/******************************************************************************
* $Revision:   1.1  $  $Author:   bcoulam  $
* $Modtime:   Jul 29 2002 12:18:28  $
*
******************************************************************************/
AS

/* Package global variables, exceptions, etc. -------------------------------*/
/* Private procedures and functions -----------------------------------------*/
/* Public procedure and function definitions --------------------------------*/
-------------------------------------------------------------------------------
PROCEDURE init_counters
(
   ior_sync  IN OUT sync%ROWTYPE
)
IS
BEGIN
   -- the caller should have already filled the critical fields
   SELECT sync_id_seq.NEXTVAL INTO ior_sync.sync_id FROM dual;
   ior_sync.num_opened := 0;
   ior_sync.num_modified := 0;
   ior_sync.num_cleared := 0;
END init_counters;

-------------------------------------------------------------------------------
PROCEDURE ins
(
   ir_sync  IN sync%ROWTYPE
)
IS
   PRAGMA AUTONOMOUS_TRANSACTION;
   ln_seq_id   NUMBER;
BEGIN
   IF (ir_sync.sync_id IS NOT NULL) THEN
      ln_seq_id := ir_sync.sync_id;
   ELSE
      SELECT sync_id_seq.NEXTVAL INTO ln_seq_id FROM dual;
   END IF;

   INSERT INTO sync
   (

 sync_id
   ,elmt_cd
   ,bgn_dtm
   ,end_dtm
   ,switch_dtm
   ,num_msgs
   ,num_opened
   ,num_modified
   ,num_cleared
   )
   VALUES
   (
    ln_seq_id
   ,ir_sync.elmt_cd
   ,ir_sync.bgn_dtm
   ,ir_sync.end_dtm
   ,ir_sync.switch_dtm
   ,ir_sync.num_msgs
   ,ir_sync.num_opened
   ,ir_sync.num_modified
   ,ir_sync.num_cleared
   );

   IF (SQL%FOUND) THEN
      COMMIT;
   ELSE
      excp.halt(env.fs_who_am_i,'Unable to insert to sync table.');
      ROLLBACK;
   END IF;
END ins;

-------------------------------------------------------------------------------
PROCEDURE upd
(
   ir_sync  IN sync%ROWTYPE
)
IS
   PRAGMA AUTONOMOUS_TRANSACTION;
BEGIN
   -- Generic update API's like this one only work for the most generic of
   -- situations and columns that are not nullable. Write independent update
   -- routines otherwise.
   UPDATE sync
   SET
    elmt_cd = NVL( ir_sync.elmt_cd, elmt_cd )
   ,bgn_dtm = NVL( ir_sync.bgn_dtm, bgn_dtm )
   ,end_dtm = NVL( ir_sync.end_dtm, end_dtm )
   ,switch_dtm = NVL( ir_sync.switch_dtm, switch_dtm )
   ,num_msgs = NVL( ir_sync.num_msgs, num_msgs )
   ,num_opened = NVL( ir_sync.num_opened, num_opened )
   ,num_modified = NVL( ir_sync.num_modified, num_modified )
   ,num_cleared = NVL( ir_sync.num_cleared, num_cleared )
   WHERE
   
sync_id = ir_sync.sync_id;

   IF (SQL%FOUND) THEN
      COMMIT;
   ELSE
      excp.halt(env.fs_who_am_i,'Unable to update sync table given ID: '||
         ir_sync.sync_id);
      ROLLBACK;
   END IF;
END upd;

-------------------------------------------------------------------------------
PROCEDURE del
(
   in_sync_id  IN sync.sync_id%TYPE
)
IS
BEGIN
   DELETE FROM sync
   WHERE sync_id = in_sync_id;

   IF (SQL%FOUND) THEN
      COMMIT;
   ELSE
      excp.halt(env.fs_who_am_i,'Unable to delete from sync table given ID: '||
         ir_sync.sync_id);
      ROLLBACK;
   END IF;
END del;


/* Package initializations (rare)  ------------------------------------------*/

END api_sync;

tt_d (body)

CREATE OR REPLACE PACKAGE BODY tt_d
/*********************************************************************
* $Revision:   1.16  $  $Author:   bcoulam  $
* $Modtime:   26 Apr 2002 14:58:46  $
*
* Description:
*  Objects which manage access to trouble ticket system data sets which don't
*  fit within the table API packages..
*
*  Note to RMOUG readers: This packag body employs collections of record to
*  maintain separation from the business and presentation layers of this
*  application. The package is over 4 years old, but the search screens work
*  just fine given our user load. If I had the time, this would be rewritten to
*  use NOCOPY, or passing the result sets back as either table functions or
*  ref cursors. I'd also replace the column-lable map with a hash table instead.
*********************************************************************/
IS

/* package global variables, cursors, etc. -------------------------*/
gar_map  tar_tt_rpt_map;
/*
The type definitions used in this package can be found in the package spec.
Since I'm not including the spec with the sample code, here are some:

TYPE tr_tt_rpt_map IS RECORD
(
   col_nm   VARCHAR2(30),
   hdr_nm   VARCHAR2(40),
   result_idx PLS_INTEGER
);

TYPE tar_tt_rpt_map IS TABLE OF tr_tt_rpt_map INDEX BY BINARY_INTEGER;

TYPE tr_tt_rpt IS RECORD
(
   tick_id   tt_tkts.tick_id%TYPE
   ,svc_type_nm  tt_svc_type.svc_type_nm%TYPE
   ,ugrp_cd   tt_tkts.ugrp_cd%TYPE
   ,assoc_cd   tt_associate.assoc_cd%TYPE
   ,status_nm   tt_status.status_nm%TYPE
   ,call_type_nm   tt_call_type.call_type_nm%TYPE
   ,prob_type_nm   tt_prob_type.prob_type_nm%TYPE
   ,author_nm   VARCHAR2(100)
   ,tick_type   tt_tkts.tick_type%TYPE
   ,priority_cd   tt_tkts.priority_cd%TYPE
   ,open_dtm   tt_tkts.open_dtm%TYPE
   ,close_dtm   tt_tkts.close_dtm%TYPE
   ,closer_nm   VARCHAR2(100)
   ,tick_num   tt_tkts.tick_num%TYPE
   ,country_nm   tt_tkts.country_nm%TYPE
   ,sub_type   tt_tkts.sub_type%TYPE
   ,trg_cd   tt_tkts.trg_cd%TYPE
   ,cycle_tm   NUMBER
   ,age   NUMBER
   ,contact_age   NUMBER
   ,parent_tick_id tt_tkts.parent_tick_id%TYPE
   ,skip_flg tt_tkts.skip_flg%TYPE
);

TYPE tar_tt_rpt IS TABLE OF tr_tt_rpt INDEX BY BINARY_INTEGER;

*/
/* private procedures and functions --------------------------------*/

/* public procedures and functions ---------------------------------*/
----------------------------------------------------------------------
PROCEDURE determine_prd_date_range
(
   is_rpt_prd IN VARCHAR2 DEFAULT NULL,
   od_bgn_dtm  OUT   DATE,
   od_end_dtm  OUT   DATE
)
IS
   ld_bgn_dtm  DATE;
   ld_end_dtm  DATE;
   ln_qrtr PLS_INTEGER;
BEGIN
   -- determine the date range for acceptible results
   IF (is_rpt_prd = '' OR is_rpt_prd IS NULL) THEN
      ld_end_dtm  := SYSDATE;
      ld_bgn_dtm  := TO_DATE('1997Jan01','YYYYMonDD'); -- date company started
   ELSIF (is_rpt_prd = '60m') THEN
      ld_end_dtm  := SYSDATE;
      ld_bgn_dtm  := ld_end_dtm - c.ONE_HOUR;
   ELSIF (is_rpt_prd = '1h') THEN
      ld_end_dtm  := utils.fd_trunc_dt(SYSDATE, 60);
      ld_bgn_dtm  := ld_end_dtm - c.ONE_HOUR;
   ELSIF (is_rpt_prd = '24h') THEN
      ld_end_dtm  := SYSDATE;
      ld_bgn_dtm  := ld_end_dtm - 1;
   ELSIF (is_rpt_prd = '1d') THEN
      ld_end_dtm  := TRUNC(SYSDATE);
      ld_bgn_dtm  := ld_end_dtm - 1;
   ELSIF (is_rpt_prd = '7d') THEN
      ld_end_dtm  := SYSDATE;
      ld_bgn_dtm  := ld_end_dtm - 7;
   ELSIF (is_rpt_prd = '1w') THEN
      ld_end_dtm  := TRUNC(SYSDATE,'D');
      ld_bgn_dtm  := ld_end_dtm - 7;
   ELSIF (is_rpt_prd = '30d') THEN
      ld_end_dtm  := SYSDATE;
      ld_bgn_dtm  := ld_end_dtm - 30;
   ELSIF (is_rpt_prd = '1M') THEN
      ld_end_dtm  := TRUNC(SYSDATE, 'Month');
      ld_bgn_dtm  := ADD_MONTHS(ld_end_dtm, -1);
   ELSIF (is_rpt_prd = 'XQ') THEN
      ld_end_dtm  := SYSDATE;
      ln_qrtr := TO_NUMBER(TO_CHAR(ld_end_dtm,'Q'));
      IF (ln_qrtr = 1) THEN
         ld_bgn_dtm  := TO_DATE(TO_CHAR(ld_end_dtm,'YYYY')||'Jan01','YYYYMonDD');
      ELSIF (ln_qrtr = 2) THEN
         ld_bgn_dtm  := TO_DATE(TO_CHAR(ld_end_dtm,'YYYY')||'Apr01','YYYYMonDD');
      ELSIF (ln_qrtr = 3) THEN
         ld_bgn_dtm  := TO_DATE(TO_CHAR(ld_end_dtm,'YYYY')||'Jul01','YYYYMonDD');
      ELSIF (ln_qrtr = 4) THEN
         ld_bgn_dtm  := TO_DATE(TO_CHAR(ld_end_dtm,'YYYY')||'Oct01','YYYYMonDD');
      END IF;
   ELSIF (is_rpt_prd = '1Q') THEN
      ln_qrtr := TO_NUMBER(TO_CHAR(SYSDATE,'Q'));
      IF (ln_qrtr = 1) THEN
         -- set up date to encompass the 4th quarter
         ld_end_dtm  := TO_DATE(TO_CHAR(SYSDATE,'YYYY')||'Jan01','YYYYMonDD')-1;
         ld_bgn_dtm  := TO_DATE(TO_CHAR(TRUNC(SYSDATE,'Y')-1,'YYYY')||'Oct01','YYYYMonDD');
      ELSIF (ln_qrtr = 2) THEN
         -- set up date to encompass the 1st quarter
         ld_end_dtm  := TO_DATE(TO_CHAR(SYSDATE,'YYYY')||'Apr01','YYYYMonDD')-1;
         ld_bgn_dtm  := TO_DATE(TO_CHAR(SYSDATE,'YYYY')||'Jan01','YYYYMonDD');
      ELSIF (ln_qrtr = 3) THEN
         -- set up date to encompass the 2nd quarter
         ld_end_dtm  := TO_DATE(TO_CHAR(SYSDATE,'YYYY')||'Jul01','YYYYMonDD')-1;
         ld_bgn_dtm  := TO_DATE(TO_CHAR(SYSDATE,'YYYY')||'Apr01','YYYYMonDD');
      ELSIF (ln_qrtr = 4) THEN
         -- set up date to encompass the 3rd quarter
         ld_end_dtm  := TO_DATE(TO_CHAR(SYSDATE,'YYYY')||'Oct01','YYYYMonDD')-1;
         ld_bgn_dtm  := TO_DATE(TO_CHAR(SYSDATE,'YYYY')||'Jul01','YYYYMonDD');
      END IF;
   END IF;

   od_bgn_dtm := ld_bgn_dtm;
   od_end_dtm := ld_end_dtm;

END determine_prd_date_range;

----------------------------------------------------------------------
FUNCTION far_quicksearch
(
   in_tick_id  IN tt_tkts.tick_id%TYPE,
   is_group_cd IN w_ugrp.ugrp_cd%TYPE,
   is_svc_type_id  IN tt_tkts.svc_type_id%TYPE,
   in_status_id IN tt_status.status_id%TYPE,
   is_skip_flg IN tt_tkts.skip_flg%TYPE,
   is_rpt_prd IN VARCHAR2 DEFAULT NULL
) RETURN tar_tt_quicksearch
IS
   CURSOR cur_tt IS
   SELECT
      tt.tick_id
      ,DECODE(ol.locked_obj_id,NULL,'N','Y') locked
      ,ts.svc_type_nm
      ,tt.priority_cd
      ,tt.open_dtm
      ,a.assoc_cd
      ,g.ugrp_desc
      ,u.user_nm
      ,tt.tick_type
      ,s.status_nm
      ,p.prob_type_nm
      ,ct.call_type_nm
      ,tt.country_nm
      ,tt.trg_cd
      ,tt.close_dtm
      ,ROUND((SYSDATE - tt.open_dtm)*24) age -- rounded in hours
      ,ROUND((SYSDATE - tt.last_contact_dtm)*24) contact_age -- rounded in hours
      ,tt.skip_flg
   FROM
      tt_tkts tt
      ,tt_associate a
      ,w_user u
      ,tt_call_type ct
      ,tt_prob_type p
      ,w_ugrp g
      ,tt_status s
      ,tt_svc_type ts
      ,objlock ol
   WHERE a.assoc_id = tt.assoc_id
   AND   tt.author_nid = u.nid (+)
   AND   tt.call_type_id = ct.call_type_id
   AND   tt.prob_type_id = p.prob_type_id
   AND   tt.ugrp_cd = g.ugrp_cd
   AND   tt.status_id = s.status_id
   AND   ts.svc_type_id = tt.svc_type_id
   AND   tt.tick_id = ol.locked_obj_id (+)
   AND   (
            (in_tick_id IS NULL)
            OR
            (tt.tick_id = in_tick_id)
         )
   AND
         (
            (is_group_cd IS NULL)
            OR
            (g.ugrp_cd = is_group_cd)
         )
   AND   (
            (is_svc_type_id IS NULL)
            OR
            (tt.svc_type_id = is_svc_type_id)
         )
   AND   (
            (in_status_id IS NULL)
            OR
            (in_status_id = -1 -- Asking for all statuses EXCEPT "open"
               AND
               (s.status_id IN
                  (SELECT status_id FROM tt_status WHERE status_nm != 'CLOSED')
               )
            )
            OR
            (s.status_id =  in_status_id)
         )
   AND   (
            (is_skip_flg IS NULL)
            OR
            (tt.skip_flg = is_skip_flg)
         )
   ORDER BY
      tt.tick_id;


   lar_tt_quicksearch   tar_tt_quicksearch;
   ln_idx PLS_INTEGER := 0;
   ld_bgn_dtm  DATE;
   ld_end_dtm  DATE;
BEGIN
   IF (in_tick_id IS NOT NULL) THEN
      -- this is the temporary solution. The permanent one should be a search
      -- option on the ticket id that allows things like "> 3000" or "2300-2400"
      BEGIN
         SELECT
            tt.tick_id
            ,DECODE(ol.locked_obj_id,NULL,'N','Y') locked
            ,ts.svc_type_nm
            ,tt.priority_cd
            ,tt.open_dtm
            ,a.assoc_cd
            ,g.ugrp_desc
            ,u.user_nm
            ,tt.tick_type
            ,s.status_nm
            ,p.prob_type_nm
            ,ct.call_type_nm
            ,tt.country_nm
            ,tt.trg_cd
            ,ROUND((SYSDATE - tt.open_dtm)*24) -- rounded in hours
            ,ROUND((SYSDATE - tt.last_contact_dtm)*24) -- rounded in hours



   ,tt.skip_flg
         INTO
             lar_tt_quicksearch(1).tick_id
            ,lar_tt_quicksearch(1).locked
            ,lar_tt_quicksearch(1).svc_type_nm
            ,lar_tt_quicksearch(1).priority_cd
            ,lar_tt_quicksearch(1).open_dtm
            ,lar_tt_quicksearch(1).assoc_cd
            ,lar_tt_quicksearch(1).ugrp_desc
            ,lar_tt_quicksearch(1).user_nm
            ,lar_tt_quicksearch(1).tick_type
            ,lar_tt_quicksearch(1).status_nm
            ,lar_tt_quicksearch(1).prob_type_nm
            ,lar_tt_quicksearch(1).call_type_nm
            ,lar_tt_quicksearch(1).country_nm
            ,lar_tt_quicksearch(1).trg_cd
            ,lar_tt_quicksearch(1).age
            ,lar_tt_quicksearch(1).contact_age
            ,lar_tt_quicksearch(1).skip_flg
         FROM
            tt_tkts tt
            ,tt_associate a
            ,w_user u
            ,tt_call_type ct
            ,tt_prob_type p
            ,w_ugrp g
            ,tt_status s
            ,tt_svc_type ts
            ,objlock ol
         WHERE a.assoc_id = tt.assoc_id
         AND   tt.author_nid = u.nid (+)
         AND   tt.call_type_id = ct.call_type_id
         AND   tt.prob_type_id = p.prob_type_id
         AND   tt.ugrp_cd = g.ugrp_cd
         AND   tt.status_id = s.status_id
         AND   tt.tick_id = in_tick_id
         AND   ts.svc_type_id = tt.svc_type_id
         AND   tt.tick_id = ol.locked_obj_id (+);

      EXCEPTION
         WHEN NO_DATA_FOUND THEN
           NULL;
      END;

   ELSE -- unbounded by ticket id, use the other parameters to search

      -- first determine the begin and end date for this query.
      determine_prd_date_range(is_rpt_prd, ld_bgn_dtm, ld_end_dtm);

      -- open cursor and fill array of record with result set
      FOR lr_tt IN cur_tt LOOP
          -- the following lookup could use some performance enhancement, like
          -- a PL/SQL table that is filled upon package initialization
         IF (tt_lookup.fs_get_status_nm(in_status_id) = 'CLOSED') THEN
            -- work off the closed dtm
            IF (lr_tt.close_dtm BETWEEN ld_bgn_dtm AND ld_end_dtm) THEN
               ln_idx := ln_idx + 1;
               lar_tt_quicksearch(ln_idx).tick_id := lr_tt.tick_id;
               lar_tt_quicksearch(ln_idx).locked := lr_tt.locked;
               lar_tt_quicksearch(ln_idx).svc_type_nm := lr_tt.svc_type_nm;
               lar_tt_quicksearch(ln_idx).priority_cd := lr_tt.priority_cd;
               lar_tt_quicksearch(ln_idx).open_dtm := lr_tt.open_dtm;
               lar_tt_quicksearch(ln_idx).assoc_cd := lr_tt.assoc_cd;
               lar_tt_quicksearch(ln_idx).ugrp_desc := lr_tt.ugrp_desc;
               lar_tt_quicksearch(ln_idx).user_nm := lr_tt.user_nm;
               lar_tt_quicksearch(ln_idx).tick_type := lr_tt.tick_type;
               lar_tt_quicksearch(ln_idx).status_nm := lr_tt.status_nm;
               lar_tt_quicksearch(ln_idx).prob_type_nm := lr_tt.prob_type_nm;
               lar_tt_quicksearch(ln_idx).call_type_nm := lr_tt.call_type_nm;
               lar_tt_quicksearch(ln_idx).country_nm := lr_tt.country_nm;
               lar_tt_quicksearch(ln_idx).trg_cd := lr_tt.trg_cd;
               lar_tt_quicksearch(ln_idx).skip_flg := lr_tt.skip_flg;
               -- age and contact age are irrelevant when CLOSED, so we don't set them
            ELSE
               --skip this record
               NULL;
            END IF;
         ELSE
            -- work off the opened dtm
            IF (lr_tt.open_dtm BETWEEN ld_bgn_dtm AND ld_end_dtm) THEN
               ln_idx := ln_idx + 1;
               lar_tt_quicksearch(ln_idx).tick_id := lr_tt.tick_id;
               lar_tt_quicksearch(ln_idx).locked := lr_tt.locked;
               lar_tt_quicksearch(ln_idx).svc_type_nm := lr_tt.svc_type_nm;
               lar_tt_quicksearch(ln_idx).priority_cd := lr_tt.priority_cd;
               lar_tt_quicksearch(ln_idx).open_dtm := lr_tt.open_dtm;
               lar_tt_quicksearch(ln_idx).assoc_cd := lr_tt.assoc_cd;
               lar_tt_quicksearch(ln_idx).ugrp_desc := lr_tt.ugrp_desc;
               lar_tt_quicksearch(ln_idx).user_nm := lr_tt.user_nm;
               lar_tt_quicksearch(ln_idx).tick_type := lr_tt.tick_type;
               lar_tt_quicksearch(ln_idx).status_nm := lr_tt.status_nm;
               lar_tt_quicksearch(ln_idx).prob_type_nm := lr_tt.prob_type_nm;
               lar_tt_quicksearch(ln_idx).call_type_nm := lr_tt.call_type_nm;
               lar_tt_quicksearch(ln_idx).country_nm := lr_tt.country_nm;
               lar_tt_quicksearch(ln_idx).trg_cd := lr_tt.trg_cd;
               IF (lr_tt.status_nm <> 'CLOSED' ) THEN
                  lar_tt_quicksearch(ln_idx).age := lr_tt.age;
                  lar_tt_quicksearch(ln_idx).contact_age := lr_tt.contact_age;
               END IF;
               lar_tt_quicksearch(ln_idx).skip_flg := lr_tt.skip_flg;
            ELSE
               --skip this record
               NULL;
            END IF;
         END IF; -- checking the status
      END LOOP;

   END IF;

   RETURN lar_tt_quicksearch;
END far_quicksearch;
----------------------------------------------------------------------
PROCEDURE init_map
IS
   PROCEDURE set_vals
   (
      idx IN PLS_INTEGER,
      is_col IN VARCHAR2,
      is_hdr IN VARCHAR2
   ) IS
   BEGIN
      gar_map(idx).col_nm := is_col;
      gar_map(idx).hdr_nm := is_hdr;
      gar_map(idx).result_idx := idx;
   END set_vals;
BEGIN
   set_vals(1, 'tick_id', 'TT ID');
   set_vals(2, 'svc_type_nm', 'Svc Type');
   set_vals(3, 'ugrp_cd', 'Assigned');
   set_vals(4, 'assoc_cd', 'Company');
   set_vals(5, 'status_nm', 'Status');
   set_vals(6, 'call_type_nm', 'Call Type');
   set_vals(7, 'prob_type_nm', 'Prob Type');
   set_vals(8, 'author_nm', 'Author');
   set_vals(9, 'tick_type', 'TT Type');
   set_vals(10, 'priority_cd', 'priority_cd');
   set_vals(11, 'open_dtm', 'Opened');
   set_vals(12, 'close_dtm', 'Closed');
   set_vals(13, 'closer_nm', 'Closer');
   set_vals(14, 'tick_num', 'TT Number');
   set_vals(15, 'country_nm', 'Country');
   set_vals(16, 'sub_type', 'Sub Type');
   set_vals(17, 'trg_cd', 'Route');
   set_vals(18, 'cycle_tm', 'Cycle Time');
   set_vals(19, 'age', 'TT Age');
   set_vals(20, 'contact_age', 'Last Contacted');
   set_vals(21, 'parent_tick_id', 'Parent TT');
   set_vals(22, 'skip_flg', 'Skip?');

END init_map;

----------------------------------------------------------------------
FUNCTION fn_get_idx(is_col_nm IN VARCHAR2) RETURN PLS_INTEGER
IS
BEGIN
   FOR i IN 1..gar_map.COUNT LOOP
      IF (is_col_nm = gar_map(i).col_nm) THEN
         RETURN i;
      END IF;
   END LOOP;
END fn_get_idx;

----------------------------------------------------------------------
FUNCTION fs_get_hdr(is_col_nm IN VARCHAR2) RETURN VARCHAR2
IS
BEGIN
   RETURN gar_map(fn_get_idx(is_col_nm)).hdr_nm;
END fs_get_hdr;

----------------------------------------------------------------------
PROCEDURE get_tt_search_rslt
(
   ias_sort_val   IN t.vc30_table DEFAULT t.empty_vc30
   ,ias_sort_direction   IN t.vc5_table DEFAULT t.empty_vc5
   ,ias_query_item_nm IN t.vc30_table DEFAULT t.empty_vc30
   ,ias_query_op  IN t.vc30_table DEFAULT t.empty_vc30
   ,ias_query_val IN t.vc2000_table DEFAULT t.empty_vc2000
   ,is_dsql_name  IN VARCHAR2 DEFAULT NULL
   ,oar_tt_rpt  OUT   tar_tt_rpt
)
IS
   -- dynamic SQL common variables
   ls_dsql_str    VARCHAR2(32767);
   ls_dsql_select VARCHAR2(4000);
   ls_dsql_from   VARCHAR2(4000);
   ls_dsql_where VARCHAR2(10000);
   ls_dsql_order VARCHAR2(4000);
   ln_dsql_handle INTEGER;
   ln_dsql_ignore INTEGER;
   ln_idx   PLS_INTEGER := 0;

   -- dynamic SQL column variables
   ln_dsql_tick_id   tt_tkts.tick_id%TYPE;
   ls_dsql_svc_type_nm tt_svc_type.svc_type_nm%TYPE;
   ls_dsql_ugrp_cd   tt_tkts.ugrp_cd%TYPE;
   ls_dsql_assoc_cd   tt_associate.assoc_cd%TYPE;
   ls_dsql_status_nm   tt_status.status_nm%TYPE;
   ls_dsql_call_type_nm   tt_call_type.call_type_nm%TYPE;
   ls_dsql_prob_type_nm   tt_prob_type.prob_type_nm%TYPE;
   ls_dsql_author_nm   VARCHAR2(100);
   ls_dsql_tick_type   tt_tkts.tick_type%TYPE;
   ln_dsql_priority_cd   tt_tkts.priority_cd%TYPE;
   ld_dsql_open_dtm   tt_tkts.open_dtm%TYPE;
   ld_dsql_close_dtm   tt_tkts.close_dtm%TYPE;
   ls_dsql_closer_nm   VARCHAR2(100);
   ls_dsql_tick_num   tt_tkts.tick_num%TYPE;
   ls_dsql_country_nm   tt_tkts.country_nm%TYPE;
   ls_dsql_sub_type   tt_tkts.sub_type%TYPE;
   ls_dsql_trg_cd   tt_tkts.trg_cd%TYPE;
   ln_dsql_cycle_tm  NUMBER;
   ln_dsql_age   NUMBER;
   ln_dsql_contact_age   NUMBER;
--   ln_dsql_parent_tick_id tt_tkts.parent_tick_id%TYPE;
   ls_dsql_skip_flg tt_tkts.skip_flg%TYPE;
BEGIN

   -- first the basic SQL statement with parts we know we'll need
   ls_dsql_select   :=   '
      SELECT
         tick_id
         ,svc_type_nm
         ,ugrp_cd
         ,assoc_cd
         ,status_nm
         ,call_type_nm
         ,prob_type_nm
         ,author_nm
         ,tick_type
         ,priority_cd
         ,open_dtm
         ,close_dtm
         ,closer_nm
         ,tick_num
         ,country_nm
         ,sub_type
         ,trg_cd
         ,cycle_tm

      ,age

      ,contact_age

      ,skip_flg
--
      ,parent_tick_id
      ';
   ls_dsql_from := '
      FROM  vw_tick_summ
      ';

   -- now add SQL for choices the user has made (if any, a choice of '= ALL'
   -- [the default] will result in nothing being added to the WHERE

   -- SYNTAX: add_to_where(where, alias, column_nm, operator, value)
   -- NOTE: the following numbered items pertain to the order of INPUT elements on the HTML ad-hoc form
   ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(1), ias_query_op(1), TO_NUMBER(ias_query_val(1))); -- assoc_id
   ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(2), ias_query_op(2), TO_NUMBER(ias_query_val(2))); -- svc_type_id
   ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(3), ias_query_op(3), ias_query_val(3)); -- ugrp_cd
   ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(4), ias_query_op(4), ias_query_val(4)); -- priority_cd
   ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(5), ias_query_op(5), ias_query_val(5)); -- tick_type

   -- added to check for the fake "Not Closed" status
   IF (ias_query_val(6) = '-1') THEN -- Not Closed is set up in tt_dd as -1
      ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(6), '<>', 2); -- status_id
   ELSE
      ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(6), ias_query_op(6), TO_NUMBER(ias_query_val(6))); -- status_id
   END IF;
   ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(7), ias_query_op(7), UPPER(ias_query_val(7))); -- trg_cd
   ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(8), ias_query_op(8), ias_query_val(8)); -- country_nm
   ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(9), ias_query_op(9), TO_NUMBER(ias_query_val(9))); -- prob_type_id
   ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(10), ias_query_op(10), TO_NUMBER(ias_query_val(10))); -- author_nid
   ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(11), ias_query_op(11), TO_NUMBER(ias_query_val(11))); -- closer_nid
   ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(12), ias_query_op(12), TO_DATE(ias_query_val(12),ia_c.UI_DATEMASK)); -- open_dtm
   ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(13), ias_query_op(13), TO_DATE(ias_query_val(13),ia_c.UI_DATEMASK)); -- close_dtm
   ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(14), ias_query_op(14), TO_NUMBER(ias_query_val(14))); -- ticket age
   ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(15), ias_query_op(15), TO_NUMBER(ias_query_val(15))); -- last contacted age
   ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(16), ias_query_op(16), ias_query_val(16)); -- skip flag
   --   ia_sql.add_to_where(ls_dsql_where, '', ias_query_item_nm(17), ias_query_op(17), TO_NUMBER(ias_query_val(17))); -- parent ticket id

   -- now build the ORDER BY clause (if any)
   ia_sql.add_to_order(ls_dsql_order, ias_sort_direction(1), NULL, ias_sort_val(1));
   ia_sql.add_to_order(ls_dsql_order, ias_sort_direction(2), NULL, ias_sort_val(2));
   ia_sql.add_to_order(ls_dsql_order, ias_sort_direction(3), NULL, ias_sort_val(3));
   ia_sql.add_to_order(ls_dsql_order, ias_sort_direction(4), NULL, ias_sort_val(4));

   -- finally, build the whole statement
   ls_dsql_str := ls_dsql_select||ls_dsql_from||ls_dsql_where||ls_dsql_order;

   logs.ins(ls_dsql_str);

   ln_dsql_handle   :=   dbms_sql.open_cursor;

   dbms_sql.parse ( ln_dsql_handle, ls_dsql_str, dbms_sql.native );

   dbms_sql.define_column ( ln_dsql_handle, 1, ln_dsql_tick_id);
   dbms_sql.define_column ( ln_dsql_handle, 2, ls_dsql_svc_type_nm, SVC_TYPE_NM_MAXLEN);
   dbms_sql.define_column ( ln_dsql_handle, 3, ls_dsql_ugrp_cd, ia_c.UGRP_CD_MAXLEN);
   dbms_sql.define_column ( ln_dsql_handle, 4, ls_dsql_assoc_cd, ASSOC_CD_MAXLEN);
   dbms_sql.define_column ( ln_dsql_handle, 5, ls_dsql_status_nm, STATUS_NM_MAXLEN);
   dbms_sql.define_column ( ln_dsql_handle, 6, ls_dsql_call_type_nm, CALL_TYPE_NM_MAXLEN);
   dbms_sql.define_column ( ln_dsql_handle, 7, ls_dsql_prob_type_nm, PROB_TYPE_NM_MAXLEN);
   dbms_sql.define_column ( ln_dsql_handle, 8, ls_dsql_author_nm, FULL_NM_MAXLEN);
   dbms_sql.define_column ( ln_dsql_handle, 9, ls_dsql_tick_type, TICK_TYPE_MAXLEN);
   dbms_sql.define_column ( ln_dsql_handle, 10, ln_dsql_priority_cd);
   dbms_sql.define_column ( ln_dsql_handle, 11, ld_dsql_open_dtm);
   dbms_sql.define_column ( ln_dsql_handle, 12, ld_dsql_close_dtm);
   dbms_sql.define_column ( ln_dsql_handle, 13, ls_dsql_closer_nm, FULL_NM_MAXLEN);
   dbms_sql.define_column ( ln_dsql_handle, 14, ls_dsql_tick_num, TICK_NUM_MAXLEN);
   dbms_sql.define_column ( ln_dsql_handle, 15, ls_dsql_country_nm, COUNTRY_NM_MAXLEN);
   dbms_sql.define_column ( ln_dsql_handle, 16, ls_dsql_sub_type, SUB_TYPE_MAXLEN);
   dbms_sql.define_column ( ln_dsql_handle, 17, ls_dsql_trg_cd, TRG_CD_MAXLEN);
   dbms_sql.define_column ( ln_dsql_handle, 18, ln_dsql_cycle_tm);
   dbms_sql.define_column ( ln_dsql_handle, 19, ln_dsql_age);
   dbms_sql.define_column ( ln_dsql_handle, 20, ln_dsql_contact_age);
   dbms_sql.define_column ( ln_dsql_handle, 21, ls_dsql_skip_flg, 1);
--   dbms_sql.define_column ( ln_dsql_handle, 22, ln_dsql_parent_tick_id);

   ln_dsql_ignore   :=   dbms_sql.execute(ln_dsql_handle);

   LOOP
      IF (dbms_sql.fetch_rows(ln_dsql_handle) > 0) THEN
         dbms_sql.column_value ( ln_dsql_handle, 1, ln_dsql_tick_id);
         dbms_sql.column_value ( ln_dsql_handle, 2, ls_dsql_svc_type_nm);
         dbms_sql.column_value ( ln_dsql_handle, 3, ls_dsql_ugrp_cd);
         dbms_sql.column_value ( ln_dsql_handle, 4, ls_dsql_assoc_cd);
         dbms_sql.column_value ( ln_dsql_handle, 5, ls_dsql_status_nm);
         dbms_sql.column_value ( ln_dsql_handle, 6, ls_dsql_call_type_nm);
         dbms_sql.column_value ( ln_dsql_handle, 7, ls_dsql_prob_type_nm);
         dbms_sql.column_value ( ln_dsql_handle, 8, ls_dsql_author_nm);
         dbms_sql.column_value ( ln_dsql_handle, 9, ls_dsql_tick_type);
         dbms_sql.column_value ( ln_dsql_handle, 10, ln_dsql_priority_cd);
         dbms_sql.column_value ( ln_dsql_handle, 11, ld_dsql_open_dtm);
         dbms_sql.column_value ( ln_dsql_handle, 12, ld_dsql_close_dtm);
         dbms_sql.column_value ( ln_dsql_handle, 13, ls_dsql_closer_nm);
         dbms_sql.column_value ( ln_dsql_handle, 14, ls_dsql_tick_num);
         dbms_sql.column_value ( ln_dsql_handle, 15, ls_dsql_country_nm);
         dbms_sql.column_value ( ln_dsql_handle, 16, ls_dsql_sub_type);
         dbms_sql.column_value ( ln_dsql_handle, 17, ls_dsql_trg_cd);
         dbms_sql.column_value ( ln_dsql_handle, 18, ln_dsql_cycle_tm);
         dbms_sql.column_value ( ln_dsql_handle, 19, ln_dsql_age);
         dbms_sql.column_value ( ln_dsql_handle, 20, ln_dsql_contact_age);
         dbms_sql.column_value ( ln_dsql_handle, 21, ls_dsql_skip_flg);
--         dbms_sql.column_value ( ln_dsql_handle, 22, ln_dsql_parent_tick_id);

         ln_idx := ln_idx + 1;

         oar_tt_rpt(ln_idx).tick_id := ln_dsql_tick_id;
         oar_tt_rpt(ln_idx).svc_type_nm := ls_dsql_svc_type_nm;
         oar_tt_rpt(ln_idx).ugrp_cd := ls_dsql_ugrp_cd;
         oar_tt_rpt(ln_idx).assoc_cd := ls_dsql_assoc_cd;
         oar_tt_rpt(ln_idx).status_nm := ls_dsql_status_nm;
         oar_tt_rpt(ln_idx).call_type_nm := ls_dsql_call_type_nm;
         oar_tt_rpt(ln_idx).prob_type_nm := ls_dsql_prob_type_nm;
         oar_tt_rpt(ln_idx).author_nm := ls_dsql_author_nm;
         oar_tt_rpt(ln_idx).tick_type := ls_dsql_tick_type;
         oar_tt_rpt(ln_idx).priority_cd := ln_dsql_priority_cd;
         oar_tt_rpt(ln_idx).open_dtm := ld_dsql_open_dtm;
         oar_tt_rpt(ln_idx).close_dtm := ld_dsql_close_dtm;
         oar_tt_rpt(ln_idx).closer_nm := ls_dsql_closer_nm;
         oar_tt_rpt(ln_idx).tick_num := ls_dsql_tick_num;
         oar_tt_rpt(ln_idx).country_nm := ls_dsql_country_nm;
         oar_tt_rpt(ln_idx).sub_type := ls_dsql_sub_type;
         oar_tt_rpt(ln_idx).trg_cd := ls_dsql_trg_cd;
         oar_tt_rpt(ln_idx).cycle_tm := ln_dsql_cycle_tm;
         IF (ls_dsql_status_nm <> 'CLOSED' ) THEN
            oar_tt_rpt(ln_idx).age := ln_dsql_age;
            oar_tt_rpt(ln_idx).contact_age := ln_dsql_contact_age;
         END IF;
         oar_tt_rpt(ln_idx).skip_flg := ls_dsql_skip_flg;
--         oar_tt_rpt(ln_idx).parent_tick_id := ln_dsql_parent_tick_id;

      ELSE
         EXIT;
      END IF;
   END LOOP;

   dbms_sql.close_cursor( ln_dsql_handle );
END get_tt_search_rslt;

END tt_d;

